Conducting a Three-Way Independent Samples Factorial ANOVA With SAS

The program, ANOVA3.sas, is available at http://core.ecu.edu/psyc/wuenschk/SAS/SAS-Programs.htm . The data are from page 447 of the 8th edition of Howell's *Statistical Methods for Psychology*. The ANOVA factors are experience level of the driver who is being tested, type of road on which the test is given, and time of day the test is given. The outcome variable is the number of steering corrections made during the one mile test session. Selected parts of the program and most of the statistical output appear in this document.

Formatting the Values of the Factors

Since the values are numerically coded, I employed Proc Format to create formats for the ANOVA factors.

```
proc format; value EX 1='Inexperienced' 2='Experienced';
value R 1='First Class' 2='Second Class' 3='Dirt';
value T 1='Day' 2='Night';
```

Macro for Confidence Intervals (code in macros to be called later in the program)

```
%macro CIt;
d = t/sqrt(n1*n2/(n1+n2));
ncp_lower = TNONCT(t,df,.975);
ncp_upper = TNONCT(t,df,.025);
d_lower = ncp_lower*sqrt((n1+n2)/(n1*n2));
d_upper = ncp_upper*sqrt((n1+n2)/(n1*n2));
output; run; proc print; var g d_lower d_upper; run;
%mend CIt;
```

Unpacking the Data With Do Loops

To save space, I packed the data. On odd numbered data lines are values of experience, road, and time. After reading those values, SAS moves to the next line of data, on which there are the four scores for members of the group just defined. The DO loop instructs SAS to INPUT the corrections scores and output the record to the data set ‘drive’ four times The @@ prevents SAS from moving to the next data line before the DO loop has executed four times. Notice that I applied the FORMAT in the data step. CARDS is used to indicate that data lines follow.

```
DATA drive;
INPUT Experience Road Time;
DO I = 1 TO 4; INPUT Corrections @@; OUTPUT; END;
format Experience EX. Road R. Time T.;
CARDS;
1 1 1
4 18 8 10
```

The Omnibus Analysis

PROC GLM was employed, despite having equal cell sizes, because I wished to use LSMEANS. With equal cell sizes, Type I sums of squares and Type III sums of squares are identical. I specified Type I – this would not be appropriate if the design were nonorthogonal. I used the bar notation to specify a complete factorial model and to obtain all cell and marginal means. The output shows that all three main effects are significant, as is the interaction between experience and time.

```
PROC GLM; CLASS Experience Road Time;
 MODEL Corrections = Experience|Road|Time / EFFECTSIZE alpha=0.1 ss1;
 MEANS Experience|Road|Time;
 LSMEANS Experience*Time / slice = Time;
Title 'Omnibus ANOVA and Simple Effects of Experience at Each Level of Time, Pooled
Error'; run;
```

I think that it is appropriate to interpret main effects of factors that are involved in a significant, monotonic interaction, so I have done so here. From the marginal means that are given, it is apparent that inexperienced drivers made significantly more steering corrections than did experienced drivers, and that steering corrections were significantly more frequent with nocturnal driving that with diurnal driving. Although I have not done so, one might wish to conduct pairwise comparisons (Fisher's procedure) among the marginal means for type of road. I think the pattern is pretty clear without additional analysis: Corrections are less frequent on better roads.

Level of Experience	N	Corrections Mean	Corrections Std Dev
Experienced	24	12.0833333	5.9191411
Inexperienced	24	22.5000000	10.6730054

Level of Road	N	Corrections Mean	Corrections Std Dev
Dirt	16	23.1250000	11.4010233
First Class	16	11.8750000	6.6017674
Second Class	16	16.8750000	8.5936023

Level of Time	N	Corrections Mean	Corrections Std Dev
Day	24	12.9166667	6.8201534
Night	24	21.6666667	10.9133610

The ratio of the largest cell variance to smallest cell variance is 2.44, so there is not any great problem with heterogeneity of variance.

Omnibus ANOVA and Simple Effects of Experience at Each Level of Time, Pooled Error

The GLM Procedure

Class Level Information

Class	Levels	Values
Experience	2	Experienced Inexperienced
Road	3	Dirt First Class Second Class
Time	2	Day Night

Dependent Variable: Corrections

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	11	3766.916667	342.446970	12.83	<.0001
Error	36	961.000000	26.694444		
Corrected Total	47	4727.916667			

R-Square	Coeff Var	Root MSE	Corrections Mean
0.796739	29.87952	5.166667	17.29167

Source	DF	Type I SS	Mean Square	F Value	Pr > F
Experience	1	1302.083333	1302.083333	48.78	**<.0001**
Road	2	1016.666667	508.333333	19.04	**<.0001**
Experience*Road	2	116.666667	58.333333	2.19	0.1271
Time	1	918.750000	918.750000	34.42	**<.0001**
Experience*Time	1	216.750000	216.750000	8.12	**0.0072**
Road*Time	2	50.000000	25.000000	0.94	0.4013
Experience*Road*Time	2	146.000000	73.000000	2.73	0.0784

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 90% Confidence Limits	
Experience	0.2754	0.2682	0.1045	0.4248
Road	0.2150	0.2026	0.0455	0.3542
Experience*Road	0.0247	0.0133	0.0000	0.1058
Time	0.1943	0.1876	0.0490	0.3474
Experience*Time	0.0458	0.0400	0.0000	0.1707
Road*Time	0.0106	-0.0007	0.0000	0.0639
Experience*Road*Time	0.0309	0.0195	0.0000	0.1193

Notice that the Experience x Time effect has a significant *F*, but the confidence interval for η^2 includes zero. The confidence interval for partial η^2 excludes zero.

Partial Variation Accounted For

Source	Partial Eta-Square	Partial Omega-Square	90% Confidence Limits	
Experience	0.5754	0.4988	0.3261	0.6317
Road	0.5141	0.4292	0.2498	0.5776
Experience*Road	0.1083	0.0471	0.0000	0.2115
Time	0.4888	0.4104	0.2346	0.5609
Experience*Time	0.1840	0.1292	0.0244	0.3028
Road*Time	0.0495	-0.0027	0.0000	0.1370
Experience*Road*Time	0.1319	0.0674	0.0000	0.2372

The sum of the values of partial eta-square is more than 215%.

Diurnal and Nocturnal Simple Main Effects of Experience

I elected to approach the significant Experience x Time interaction by testing the diurnal and nocturnal simple main effects of experience. "LSMEANS Experience*Time / slice = Time;" tests these simple effects with pooled error.

Least Squares Means

Experience	Time	Corrections LSMEAN
Experienced	Day	9.8333333
Inexperienced	Day	16.0000000
Experienced	Night	14.3333333
Inexperienced	Night	29.0000000

Experience*Time Effect Sliced by Time for Corrections

Time	DF	Sum of Squares	Mean Square	F Value	Pr > F
Day	1	228.166667	228.166667	8.55	**0.0060**
Night	1	1290.666667	1290.666667	48.35	**<.0001**

Total Variation Accounted For

Time	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 90% Confidence Limits	
Day	0.0483	0.0424	0.0000	0.1744
Night	0.2730	0.2658	0.1027	0.4226

The eta-squared for Experience during the day is $\frac{SS_{Exp|Day}}{SS_{Total}} = \frac{228.16666}{4727.91666} = 0.048$. Notice that the denominator is the total sum of squares from the factorial model.

The eta-squared for Experience during the night is $\frac{SS_{Exp|Night}}{SS_{Total}} = \frac{1290.6666}{4727.91666} = 0.273$. Notice that the denominator is the total sum of squares from the factorial model.

Partial Variation Accounted For

Time	Partial Eta-Square	Partial Omega-Square	90% Confidence Limits	
Day	0.1919	0.1359	0.0275	0.3100
Night	0.5732	0.4966	0.3237	0.6300

I also tested these simple effects with individual error terms, sorting by time and then doing an Experience x Road ANOVA by time. Please notice that I did include both type of road and the Experience x Road interaction in the model. I included these effects in order to exclude from the error term variance accounted by them, providing a more powerful test. The plot of these simple main effects makes the interaction quite clear (especially if you take the time to draw in the lines, which I have done with Word's drawing tool): The effect of experience is much greater (its line much steeper) with nocturnal driving than with diurnal driving.

```
PROC SORT; BY Time;
PROC GLM; CLASS Experience Road;
MODEL Corrections = Experience|Road / ss1 EFFECTSIZE alpha=0.1; BY Time;
title 'Simple effects at levels of Time, Individual Error.'; run;
proc means NWAY noprint; class Experience Time; var Corrections; output out=ExT mean= ;
proc plot; plot Corrections*Experience=Time;run;
Proc gplot;
symbol1 interpol=join width=4 value=triangle height=2 color=red;
symbol2 interpol=join width=4 value=square height=2 color=gray;
plot Corrections*Experience=Time / haxis=1 to 2 by 1;
title 'Figure 1. Mean Corrections by Experience and Time'; run;
```

```
                    Simple Effects at Levels of Time, Individual Error.

------------------------------------------- Time=Day -------------------------------------------

Dependent Variable: Corrections

                                               Sum of
      Source                      DF          Squares     Mean Square    F Value    Pr > F

      Model                        5       612.833333      122.566667       4.83    0.0057

      Error                       18       457.000000       25.388889

      Corrected Total             23      1069.833333


                    R-Square     Coeff Var      Root MSE    Corrections Mean

                    0.572831      39.00959      5.038739            12.91667


      Source                      DF         Anova SS     Mean Square    F Value    Pr > F

      Experience                   1      228.1666667     228.1666667       8.99    0.0077
      Road                         2      308.3333333     154.1666667       6.07    0.0097
      Experience*Road              2       76.3333333      38.1666667       1.50    0.2490

                                      Total Variation Accounted For

                                                     Semipartial
                                        Semipartial       Omega-          Conservative
             Source                      Eta-Square       Square     90% Confidence Limits

             Experience                      0.2133       0.1851      0.0170      0.4199
             Road                            0.2882       0.2352      0.0195      0.4635
             Experience*Road                 0.0714       0.0233      0.0000      0.2264
```

The eta-squared for Experience during the day is $\frac{SS_{Exp|Day}}{SS_{Total|Day}} = \frac{228.16666}{1069.8333} = 0.213$. Notice that the denominator is the total sum of squares for the data gathered during the day, excluding those gathered at night.

Partial Variation Accounted For

Source	Partial Eta-Square	Partial Omega-Square	90% Confidence Limits	
Experience	0.3330	0.2497	0.0500	0.4895
Road	0.4029	0.2971	0.0640	0.5328
Experience*Road	0.1431	0.0403	0.0000	0.3005

--------------------------------------- Time=Night ---------------------------------------

Dependent Variable: Corrections

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	2235.333333	447.066667	15.97	<.0001
Error	18	504.000000	28.000000		
Corrected Total	23	2739.333333			

R-Square	Coeff Var	Root MSE	Corrections Mean
0.816014	24.42232	5.291503	21.66667

Source	DF	Anova SS	Mean Square	F Value	Pr > F
Experience	1	1290.666667	1290.666667	46.10	<.0001
Road	2	758.333333	379.166667	13.54	0.0003
Experience*Road	2	186.333333	93.166667	3.33	0.0589

Total Variation Accounted For

Source	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 90% Confidence Limits	
Experience	0.4712	0.4563	0.1951	0.6276
Road	0.2768	0.2538	0.0136	0.4534
Experience*Road	0.0680	0.0471	0.0000	0.2211

The eta-squared for Experience during the night is $\frac{SS_{Exp|Night}}{SS_{Total|Night}} = \frac{1290.6666}{2739.3333} = 0.471$. Notice that the denominator is the total sum of squares for the data gathered during the night, excluding those gathered at day.

Partial Variation Accounted For

Source	Partial Eta-Square	Partial Omega-Square	90% Confidence Limits	
Experience	0.7192	0.6527	0.4298	0.7791
Road	0.6007	0.5110	0.2507	0.6863
Experience*Road	0.2699	0.1624	0.0000	0.4223

Simple effects at levels of Time, Individual Error.

Plot of Corrections*Experience. Symbol is value of Time.



Figure 1. Mean Corrections by Experience and Time



If the Triple Interaction Were Significant

For pedagogical purposes, let us pretend that the triple interaction is significant. Pretending such, I elected to approach it by testing the diurnal and nocturnal simple interaction between experience and type of road. Since I had earlier included Experience x Road in the model tested by levels of time of day (individual error terms), we can just look back at that output to get the tests we need. The output provided by SAS indicates that neither simple interaction is significant, but one of them is close. Accordingly, I decided to go ahead and do the extra work to get tests done with the pooled error term from the omnibus analysis, substituting the *MSE* from the omnibus analysis for the individual error term reported by SAS. For diurnal driving, the interaction remains not significant, $F(2,36)=\frac{38.1\overline{6}}{26.69\overline{4}}=1.42$. Note that I used the PROBF function in SAS to obtain the *p* for this *F*, .255.

The nocturnal interaction has $F(2,36)=\frac{93.1\overline{6}}{26.69\overline{4}}=3.49$, *p* = .041.

```
data p;
 F_Day = 1-PROBF(1.42, 2, 36); F_Night = 1-PROBF(3.49, 2, 36);
proc print; Title 'P values for Simple Interactions using pooled error'; run;
```

```
P values for Simple Interactions using pooled error

Obs      F_Day       F_Night

 1      0.25493     0.041177
```

Although I obtained the diurnal interaction plot, there is not much to be said about it. Since that interaction was not significant, the slopes of the three lines there do not differ from one another significantly, even though the slope with first class roads appears to be distinctly less than that with the poorer roads.

The slopes in the nocturnal interaction plot (22.5, 11.5, 10) differ by more than do those in the diurnal plot (5, 11, 2.5) -- the variance of the slopes is about 47 in the nocturnal plot and about 19 in the diurnal plot. I included the VAXIS option to control where the tick marks would appear on the ordinate. If I had not done so, the two plots would have been scaled differently, and the result would be that the diurnal interaction would appear to be greater than the nocturnal interaction. The nocturnal plot makes it pretty clear that the interaction here is a matter of the effect of experience being much greater at night on dirt roads than it is on better roads.

```
title 'Simple interactions at levels of Time.'; run;
proc means data=drive NWAY noprint; class Experience Road; var Corrections;
 output out=ExR mean= ; by Time;
proc plot; plot Corrections*Experience=Road / VAXIS=5 10 15 20 25 30 35 40;
 by Time; run;
Proc gplot;
symbol1 interpol=join width=4 value=triangle height=2 color=red;
symbol2 interpol=join width=4 value=square height=2 color=green;
symbol3 interpol=join width=4 value=circle height=2 color=brown;
By Time;
plot Corrections*Experience=Road / haxis=1 to 2 by 1;
title 'Figure 2. Corrections by Experience, Road, & Time'; run;
```

Simple interactions at levels of Time.

-- Time=Day --



Simple interactions at levels of Time.

-- Time=Night --

Plot of Corrections*Experience. Symbol is value of Road.



Figure 2. Corrections by Experience, Road, & Time



I tested the simple, simple main effects of experience for the three types of road at night by first creating a new data set, containing only the nocturnal data, and then testing the effect of experience by type of road. As you can see, the effect is significant at each level of the road variable, but is much stronger with the dirt road. I would recommend estimating values of *d* for each of these simple, simple main effects, and obtaining confidence intervals for those estimates, but I have not done that here.

```
data night; set drive; if Time = 2;  proc sort; by Road;
proc GLM; class Experience;
model Corrections = Experience / EFFECTSIZE alpha=0.1; by Road;
title 'Simple, simple main effects of Experience at Road for Time = night.'; run;
```

```
            Simple, simple main effects of Experience at Road for Time = night.

--------------------------------------- Road=First Class ---------------------------------------

                                      The ANOVA Procedure

Dependent Variable: Corrections
```

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	200.0000000	200.0000000	8.82	0.0249

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Error	6	136.0000000	22.6666667		
Corrected Total	7	336.0000000			

R-Square	Coeff Var	Root MSE	Corrections Mean
0.595238	31.73968	4.760952	15.00000

------------------------------------ Road=Second Class ------------------------------------

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	264.5000000	264.5000000	9.07	0.0237
Error	6	175.0000000	29.1666667		
Corrected Total	7	439.5000000			

R-Square	Coeff Var	Root MSE	Corrections Mean
0.601820	25.41467	5.400617	21.25000

-- Road=Dirt --

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	1012.500000	1012.500000	31.48	0.0014
Error	6	193.000000	32.166667		
Corrected Total	7	1205.500000			

R-Square	Coeff Var	Root MSE	Corrections Mean
0.839900	19.72719	5.671567	28.75000

Strength of Effect Estimates: *d*

I prefer estimated *d* over η^2 as a strength of effect estimate, so I have obtained estimated *d* for each of the one-*df* effects that was significant in the omnibus analysis and have also obtained the corresponding confidence intervals. I also obtained these same statistics for the simple main effects of experience, day and night.

```
title 'Get t values for estimating d with confidence interval.'; run;
proc ttest data=drive; class experience; var corrections;
proc ttest data=drive; class time; var corrections;
proc ttest data=drive; class experience; var corrections; by time; run;
***********************************************************************;
title 'Estimated d for Experience.';run;
Data CId; t= 4.18; df = 46; n1 = 24; n2 = 24; %CIt
***********************************************************************;
title 'Estimated d for Time of Day.';run;
Data CId; t= 3.33; df = 46; n1 = 24; n2 = 24; %CIt
***********************************************************************;
```

```
title 'Estimated d for Experience during the Day.';run;
Data CId; t= 2.44; df =22; n1 = 12; n2 = 12;  %CIt
************************************************************************;
title 'Estimated d for Experience at Night.';run;
Data CId; t= 4.43; df =22; n1 = 12; n2 = 12; %CIt
```

Get t values for estimating d with confidence interval.

Experience	Method	Mean	95% CL Mean		Std Dev	95% CL Std Dev	
Inexperienced		22.5000	17.9932	27.0068	10.6730	8.2952	14.9717
Experienced		12.0833	9.5839	14.5828	5.9191	4.6004	8.3031
Diff (1-2)	Pooled	10.4167	5.4021	15.4312	8.6299	7.1712	10.8390

Method	Variances	DF	t Value	Pr > |t|
Pooled	Equal	46	4.18	0.0001

Get t values for estimating d with confidence interval.

Time	Method	Mean	95% CL Mean		Std Dev	95% CL Std Dev	
Day		12.9167	10.0368	15.7966	6.8202	5.3007	9.5670
Night		21.6667	17.0584	26.2750	10.9134	8.4820	15.3088
Diff (1-2)	Pooled	-8.7500	-14.0377	-3.4623	9.0999	7.5618	11.4293

Method	Variances	DF	t Value	Pr > |t|
Pooled	Equal	46	-3.33	0.0017

Simple Main Effects Follow

-- Time=Day --

Experience	Method	Mean	95% CL Mean		Std Dev	95% CL Std Dev	
Inexperienced		16.0000	11.6907	20.3093	6.7823	4.8046	11.5156
Experienced		9.8333	6.3235	13.3431	5.5241	3.9132	9.3792
Diff (1-2)	Pooled	6.1667	0.9299	11.4035	6.1853	4.7837	8.7543

Method	Variances	DF	t Value	Pr > |t|
Pooled	Equal	22	2.44	0.0231

-- Time=Night --

Experience	Method	Mean	95% CL Mean		Std Dev	95% CL Std Dev	
Inexperienced		29.0000	22.6463	35.3537	10.0000	7.0840	16.9788
Experienced		14.3333	10.7562	17.9105	5.6300	3.9883	9.5591
Diff (1-2)	Pooled	14.6667	7.7963	21.5370	8.1147	6.2759	11.4852

Method	Variances	DF	t Value	Pr > |t|
Pooled	Equal	22	4.43	0.0002

```
Estimated d for Experience.

Obs         d        d_lower     d_upper

 1       1.20666     0.58421     1.81803
```

```
Estimated d for Time of Day.

Obs         d        d_lower     d_upper

 1       0.96129     0.35787     1.55536
```

```
Estimated d for Experience during the Day.

Obs         d        d_lower     d_upper

 1       0.99613     0.13419     1.83814
```

```
Estimated d for Experience at Night.

Obs         d        d_lower     d_upper

 1       1.80854     0.83379     2.75463
```

Plots

Notice that I used both Proc Plot and Proc Gplot to produce the plots. The plots produced by Gplot can be exported to image files, just select the plot and then click "File," "Export as Image." Those produced by Proc Gplot look a lot better than those produced by Proc Plot, but Gplot can be confusing to use. A good alternative is to use Excel.

Presentation of the Results

As shown in Table 1, all of the main effects were large and significant. Experienced drivers made significantly fewer errors (M = 12.1) than did inexperienced drivers (M = 22.5), d = 1.207, 95% CI [.584, 1.818]. Errors were most frequent on dirt roads (M = 23.12), intermediate on second class roads (M = 16.88), and least frequent on first class roads (M = 11.88). Errors were significantly more frequent at night (M = 21.7) than during the day (M = 12.9), d = .961, 95% CI [.358, 1.555].

Table 1. ANOVA Source Table.

Source	*df*	*F*	*p*	η^2	$CI_{.90}$ for η^2
Experience	1	48.78	< .001	.275	.104, .425
Road	2	19.04	< .001	.215	.046, .354
Time	1	34.42	< .001	.194	.049, .347
Exp x Road	2	2.19	.12	.025	.000, .106
Exp x Time	1	8.12	.007	.046	.000, .171
Road x Time	2	0.94	.40	.011	.000 .064
E x R x T	2	2.73	.08	.031	.000, .119
Error[1]	36				

[1]*MSE* = 26.69

Among the interactions, only Experience x Time was statistically significant, and it was small in magnitude when compared to the main effects. Simple effects analysis showed that experienced drivers made significantly fewer errors than did inexperienced drivers both during the day (M = 9.83, 16.00), $F(1, 36) = 8.55$, $p = .006$, $d = .996$, 95% CI [.134, 1.838], and at night (M = 14.33, 29.00) , $F(1, 36) = 48.35$, $p < .001$, $d = 1.809$, 95% CI [.834, 2.755], with the difference at night greater than during the day (see Figure 1).

Figure 1. Mean Corrections by Experience and Time



For Even More Fun, Try ….

Wow, three-way ANOVA certainly is a lot more fun than one- or two-way ANOVA. Now I want to do a six-way ANOVA. With how many *F* tests would I be blessed with six factors? The answer is here.

Karl L. Wuensch, Dept. of Psychology, East Carolina Univ., Greenville, NC, USA
February, 2018.

SAS Lessons